

## NOTA DE APLICACION 2.

### Leyendo el valor de una entrada analógica.

#### Breve descripción de la aplicación.

La conversión análogo – digital constituye una tarea presente hoy en día en muchas aplicaciones desarrolladas a microcontroladores que son empleadas tanto en la instrumentación, el control, las comunicaciones, la bioingeniería, etc. Este proceso consiste en la obtención de un valor digital, representativo del valor, por lo general del voltaje, de una señal analógica en un instante de tiempo dado.

Para realizar esta tarea se diseñan hoy en día circuitos integrados conocidos con el nombre de conversores análogo – digitales cuyas características fundamentales son:


**numero de bits.** Es la cantidad de bits que emplea el conversor para representar el valor de la señal analógica. Este parámetro está estrechamente relacionado con la resolución del conversor, a mayor número de bits mayor resolución.

**tiempo de conversión.** Es el tiempo que toma al conversor realizar la operación de conversión a partir del momento en que se le da la orden de convertir. Está muy relacionado con la técnica de conversión implementada en el conversor y debe ser tomada en cuenta en dependencia de la aplicación que se está resolviendo. Por regla general a menor tiempo de conversión, mayor costo.

En esta Nota de Aplicación se conecta el conversor análogo – digital ADC 0801 de la Analog Device a un microcontrolador 8051 para tomar el valor del voltaje que se cae en uno de los brazos de un divisor resistivo construido con un potenciómetro lineal. El voltaje a convertir es aplicado entre los terminales Vin+ y Vin- del conversor al que se le da inicio de conversión poniendo a 0 su terminal WR/ y es leído poniendo a 0 su terminal RD/. El conversor indica que ha terminado la conversión llevando a 0 su terminal INT/ el que es atendido por interrupción para lo cual se emplea la entrada de interrupción EXTERNA 0 del microcontrolador.

El valor del voltaje es mostrado a través de un display conformado por dos lámparas de 7 segmentos del tipo ánodo común y el mismo es mostrado en hexadecimal. La orden de tomar un valor es dada mediante el apretado de una Tecla conectada al terminal P3.0 del microcontrolador y atendida por encuesta. Las resistencias conectadas entre los terminales de salida del microcontrolador y las entradas de segmentos de las lámparas fueron calculadas para que la corriente que circulara por cada segmento cuando es encendido sea de 10 miliamperes.

## Hardware.



## Software.

```

; ***** Leyendo una entrada analógica *****
; ///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
; El convertor análogo digital utilizado es el convertor ADC0801 de la Analog
; Device de 8 bits y de aproximaciones sucesivas.
; La señal analógica que se esta leyendo es el voltaje resultante de un divisor de
; voltaje hecho con un potenciómetro y la Fuente Vcc tomado por el convertor
; entre sus terminales Vin+ y Vin-.
; El convertor empleado tiene tres entradas de control,
;   CS/      actúa como una entrada de selección. Activa en 0.
;   WR/      para dar Inicio de Conversión. Activa en 0.
;   RD/      para leer el resultado de la conversión. Activa en 0.
; El convertor indica Fin de Conversión poniendo a 0 su terminal INT/.
; El Fin de Conversión es atendido por interrupción empleando para ello la
; entrada INT0/ del microcontrolador.
; El dato resultante de la conversión es entregado por los terminales DB0 a DB7.
; En el display se muestra el valor en hexadecimal leído del convertor.
; ///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
; Inicio del Programa.
    org 0
    sjmp inicio ;salto al inicio del programa principal.
; ///////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
; Subrutina de atención a la interrupción EXTERNA 0 del microcontrolador
; empleada para atender el Fin de Conversión.

```

```

org      3
mov      IE,#10000000b    ;deshabilitando la interrupción EXTERNA 0
clr      P3.5    ;dando lectura al ADC0801, CS/ esta a 0.
nop      ;esperando por la respuesta del conversor   ADC0801
nop
mov      A,P1    ;leyendo el ADC0801
mov      R7,A    ;salvando el dato leído
setb     20h.0    ;indicando que hay un dato leído listo para ser mostrado
setb     P3.5    ;quitando orden de lectura del ADC0801
reti
; ////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
; Inicio del programa principal.
inicio:   clr      20h.0    ;limpiando bandera de dato leído
          mov      R7,#0    ;limpiando el registro de dato leído
          setb     TCON.0    ;interrupción EXTERNA 0 activa por frente
; Indicando a través del display que el equipo está encendido y listo para trabajar
          mov      P0,#0BFH ;escribe el guion en la lámpara 2.
          mov      P2,#0BFH ;escribe el guion en la lámpara 1.
          call     delay    ;demora de dos segundos.
; Apagando el display.
          mov      P0,#0FFH  ; todos los segmentos en 1.
          mov      P2,#0FFH  ; todos los segmentos en 1.
; Dando una falsa lectura al conversor.
; Esto hay que hacerlo solo una vez y con ello garantizamos que el conversor
; este listo para el trabajo. Es como un RESET al conversor.
          clr      P3.7    ; se pone SELECT a 0 y no se cambia mas
          clr      P3.5    ; LECT a 0, lectura del conversor.
          nop
          nop
          setb     P3.5
; Habilitar la interrupcion.
lazo:     clr      IE0      ;limpiando la bandera de interrupción EXTERNA 0
          mov      IE,#10000001b    ;habilitando la interrupción EXTERNA 0
; Esperando por la orden de medir el voltaje mediante el apretado de la Tecla.
          jb       P3.0,$    ;espera por tecla apretada
          jnb      P3.0,$    ;espera porque liberen la tecla
; Dando la orden de inicio de conversión.
          clr      P3.6    ; ICON a 0, inicio de conversión.
          nop
          nop
          setb     P3.6
; Esperando por dato leído, respuesta de la Subrutina de atención a Fin de
; Conversión.
          jnb      20H.0,$
          clr      20H.0    ;limpiando la bandera de dato leído

```

```

; Mostrando el dato leído en el display. El dato a mostrar esta en el Registro R7.
    mov  DPTR, #Tabla      ; puntero Tabla de códigos de lamp. 7 segm.
    mov  A,R7              ; dato amostrar al acumulador
    anl  A,#0Fh            ; digito hexadecimal menos significativo
    movc A,@A+DPTR         ; extrayendo el código de la lámpara
    mov  P2,A              ; mostrando el digito en el display
    mov  A,R7
    anl  A,#0F0h           ; digito hexadecimal mas significativo
    mov  B,#16
    div  AB                 ; rotandolo hacia parte baja de acumulador
    movc A,@A+DPTR         ; extrayendo el código de la lámpara
    mov  P0,A              ; mostrando el digito en el display
; Manteniendo el dato en el display por 2 segundos
    call delay
; Limpiando el display
    mov  P0,#0FFh          ; todos los segmentos a 1.
    mov  P2,#0FFh          ; todos los segmentos a 1.
; regresar para esperar porque aprieten nuevamente la Tecla.
    sjmp lazo              ; ciclo repetitivo.
; ///////////////////////////////////////////////////////////////////
; Subrutina que provoca una demora de 2 segundos.
; Frecuencia del reloj del micro 12 MHz
; tiempo de demora = 200 x 100 x 100 x 1 microsegundo
delay:    mov  R6,#200
lazo2:    mov  R5,#100
lazo1:    mov  R4,#100
          djnz R4,$
          djnz R5,lazo1
          djnz R6,lazo2
          ret
; ///////////////////////////////////////////////////////////////////
; Tabla con los códigos de Lámparas de 7 Segmentos de los dígitos
; hexadecimales para una Lámpara de Ánodo Común.
Tabla:    db   40h      ; digito hexadecimal 0.
          db   79h      ; digito hexadecimal 1.
          db   24h      ; digito hexadecimal 2.
          db   30h      ; digito hexadecimal 3.
          db   19h      ; digito hexadecimal 4.
          db   12h      ; digito hexadecimal 5.
          db   02h      ; digito hexadecimal 6.
          db   38h      ; digito hexadecimal 7.
          db   00h      ; digito hexadecimal 8.
          db   10h      ; digito hexadecimal 9.
          db   77h      ; digito hexadecimal A.
          db   7Ch      ; digito hexadecimal B.

```

```
db    46h    ; digito hexadecimal C.
db    21h    ; digito hexadecimal D.
db    07h    ; digito hexadecimal E.
db    0Eh    ; digito hexadecimal F.
; //////////////////////////////////////
end
; *****Fin del Programa de la Aplicación*****
```